Symbolic SPICE

Symbolic SPICE™

Circuit Analyzer and Approximator

Application Note

AN-004:

An Op-Amp Phase Shift Oscillator

by Gregory M. Wierzba

Rev 072010

A) Introduction

The schematic shown in Fig. 1 is that of an op-amp phase shift oscillator [1]. Oscillators consist of an amplifier and passive network. In this case the passive network contains three capacitors which can provide a phase shift approaching 270°. At one frequency the phase shift of the passive network is exactly 180° and this is the frequency of oscillation. For the circuit to sustain the oscillation at node 1, the inverting amplifier must boost the signal by the attenuation created by the passive network as well as to provide the necessary 180° of phase shift to return the signal to node 1 unchanged. So there are two conditions that need to be found to characterize this oscillator. These are the frequency of oscillation and the gain needed by the amplifier for oscillation.

A Symbolic SPICE™ input file, **`pso.cir`**, is given in Table 1. This file uses Symbolic SPICE's definition [2] of an ideal op-amp which begins with the letters `XOA`.



Figure 1. Phase shift oscillator

Table 1. Symbolic SPICE input file

```
Phase Shift Oscillator
R 2 0
R 3 0
R 4 5
R2 5 1
C 1 2
C 2 3
C 3 4
XOA 0 5 1
.END
```

B) Running Symbolic SPICE

Running [2] the input file shown in Table 1, the following are the prompts. The user responses are shown in bold:

```
Symbolic SPICE - Circuit Analyzer and Approximator
Demo Version 3.1
(C) Copyright 2010 by Willow Electronics, Inc.
```

INPUT FILE NAME [.cir] : **pso**

OUTPUT FILENAME [pso.det] : **(hit enter)**

Determinant string sorted according to orders of some variable? (y/n) : **n**

Numerical evaluation of the results? (y/n) : **n**

Discard terms if their magnitude falls below a threshold? (y/n) : **n**

Check and solve for second order filter functions? (y/n) : **n**

Solve for a variable or expression? (y/n) : **y**

Available Unknowns:
V1 V2 V3 V4

*Ignore nodes 6 and higher if present. They are used for internal numbering.

Valid Operators: +, -, *, /, ( ), { }, [ ]

Equation: **v1**

Solve for another variable or expression? (y/n) : **n**

C) Symbolic SPICE Determinant Output

The output file **pso.det** listed in Table 2 is the matrix written by Symbolic SPICE and the transfer function requested.

Table 2. Symbolic SPICE Output File **pso.det**

```
Phase Shift Oscillator

[0  ] [-sC          sC+sC+G      -sC          0            ][V1 ]
[0  ]=[0            -sC          sC+sC+G      -sC          ][V2 ]
[0  ] [0            0            -sC          sC+G         ][V3 ]
[0  ] [-G2          0            0            -G           ][V4 ]

*Ignore nodes 6 and higher if present. They are used for internal numbering.

Numerator of: v1

TERMS SORTED ACCORDING TO POWERS OF s

s**0 terms:

 + 0

************************************************
```

```
Denominator of: v1

TERMS SORTED ACCORDING TO POWERS OF s

s**3 terms:

 + sC*sC*sC*G2 + sC*sC*sC*G

s**2 terms:

 +  6*sC*sC*G2*G

s**1 terms:

 +  5*sC*G2*G*G

s**0 terms:

 + G2*G*G*G
***********************************************
```

Symbolic SPICE's format is a collection of strings of symbols. This is usually not how most people view equations, so you may need to do some minor editing. Thus we have:

$$\text{Denominator of v1} = s^3C^3(G_2 + G) + s^2 6C^2G_2G + s5CG_2G^2 + G_2G^3$$

Note that V_1 = `0/Denominator of v1`. This will be true for every node voltage in an oscillator because there is no input voltage. If this circuit oscillates then node voltage one is not zero. This implies that the `Denominator of v1`, which is also the characteristic equation, must be zero since 0 / 0 is indeterminate and can take on any value. If the circuit is oscillating at a single frequency then $s = j\omega$. We can use this to find the frequency of oscillation and the gain needed for oscillation as follows.

Since $s = j\omega$, then $s^2 = j\omega \bullet j\omega = -\omega^2$ and $s^3 = j\omega \bullet j\omega \bullet j\omega = -j\omega^3$. Setting the `Denominator of v1` $= 0 + j0$, we have

$$-j\omega^3C^3(G_2 + G) - \omega^2 6C^2G_2G + j\omega 5CG_2G^2 + G_2G^3 = 0 + j0 \tag{1}$$

Setting the real part of the left side of Eqn. 1 equal to the real part of the right side of Eqn. 1,

$$-\omega^2 6C^2G_2G + G_2G^3 = 0$$

$$\omega^2 6C^2G_2G = G_2G^3$$

$$\omega^2 = \frac{1}{6C^2R^2}$$

$$\omega \triangleq \omega_o = \frac{1}{\sqrt{6}RC}$$

Setting the imaginary part of the left side of Eqn. 1 equal to the imaginary part of the right side of Eqn. 1,

$$-j\omega^3 C^3 (G_2 + G) + j\omega 5 C G_2 G^2 = j0$$

$$\omega^3 C^3 (G_2 + G) = \omega 5 C G_2 G^2$$

$$\frac{G_2 + G}{G_2} = 1 + \frac{G}{G_2} = \frac{\omega 5 C G^2}{\omega^3 C^3} = \frac{5}{\omega^2 C^2 R^2} = \frac{5}{1/6} = 30$$

$$\frac{R_2}{R} = 29$$

Thus the conditions for oscillation are

$$R_2 = 29R \quad \text{and} \quad \omega_o = \frac{1}{\sqrt{6}RC} \qquad \textbf{(2)}$$

D) Phase Shift Oscillator Design

Suppose we design an oscillator for testing audio equipment. This type of testing is usually done at 1k Hz. Let C be a standard capacitance value of 6200p F then from Eqn. 2, we have

$$R = \frac{1}{6200\text{p} \ \sqrt{6} \ 2\pi \ 1000} = 10.48\text{k } \Omega$$

The nearest standard value of resistor is 10k Ω. Also from Eqn. 2, we have that

$$R_2 = 29 \bullet 10\text{k} = 290\text{k } \Omega$$

We will need a pot to adjust R_2 to the exact value needed for oscillation. Simulating this in PSpice has some problems in that a transient response rich in harmonics is needed to initiate the oscillation. One way to "kick start" the oscillator is to turn on the power with a switch. We will also need an op-amp model that includes the non-ideal effect of power supply clipping, otherwise our oscillator will produce very large voltages. A SPICE macromodel [3] will satisfy this. An LF411 low distortion op-amp was selected for the inverting amplifier. The input file is listed in Table 3.

Table 3. PSpice input file **`audio_osc.cir`**

```
Audio Oscillator
R3 2 0 10K
R4 3 0 10K
R1 4 5 10K
R2 5 1 290K
C1 1 2 6.2N
C2 2 3 6.2N
C3 3 4 6.2N
XOA 0 5 6 7 1 LF411
VCC 6 0 PULSE (0 15 1N)
VEE 7 0 -15
* connections:   non-inverting input
*                |  inverting input
*                |  |  positive power supply
*                |  |  |  negative power supply
*                |  |  |  |  output
*                |  |  |  |  |
.subckt LF411    1  2  3  4  5
*
  c1   11 12 4.196E-12
  c2    6  7 10.00E-12
  css  10 99 1.333E-12
  dc    5 53 dx
  de   54  5 dx
  dlp  90 91 dx
  dln  92 90 dx
  dp    4  3 dx
  egnd 99  0 poly(2) (3,0) (4,0) 0 .5 .5
  fb    7 99 poly(5) vb vc ve vlp vln 0 31.83E6 -30E6 30E6 30E6 -30E6
  ga    6  0 11 12 251.4E-6
  gcm   0  6 10 99 2.514E-9
  iss  10  4 dc 170.0E-6
  hlim 90  0 vlim 1K
  j1   11  2 10 jx
  j2   12  1 10 jx
  r2    6  9 100.0E3
  rd1   3 11 3.978E3
  rd2   3 12 3.978E3
  ro1   8  5 50
  ro2   7 99 25
  rp    3  4 15.00E3
  rss  10 99 1.176E6
  vb    9  0 dc 0
  vc    3 53 dc 1.500
  ve   54  4 dc 1.500
  vlim  7  8 dc 0
  vlp  91  0 dc 25
  vln   0 92 dc 25
.model dx D(Is=800.0E-18 Rs=1m)
.model jx NJF(Is=12.50E-12 Beta=743.3E-6 Vto=-1)
.ends
.PROBE
.TRAN 2.5U 5M 0 2.5U
.END
```

The schematic is shown in Fig. 2 and the output is shown in Fig. 3.



Figure 2. Phase shift audio oscillator with op-amp macromodel



Figure 3. Phase shift audio oscillator output

The period in steady-state is 955.058μ sec which is a frequency of 1047.1 Hz. Using the values of 10k Ω and 6200p F in Eqn. 2, f_o = 1047.9 Hz. This is a difference of 0.088 %.

The actual amplitude will vary from that above and depend on the actual parameters of the op-amp.

E) Conclusion

Symbolic SPICE can be used to analyze ideal op-amp circuits to find the conditions of oscillation in a phase shift oscillator. Symbolic SPICE allows identical elements to have the same SPICE name and symbolic representation. This simplifies the formulas.

F) References

[1] G. M. Wierzba, *ECE 303: Electronics Laboratory Lab Manual*, Lab VII. This e-book is available at http://stores.lulu.com/willowepublishing

[2] G. M. Wierzba, *Symbolic SPICE User Manual*. This e-book is available at http://stores.lulu.com/willowepublishing

[3] G. M. Wierzba, *ECE 402: Application of Analog ICs Class Notes,* Ch.5. This e-book is available at http://stores.lulu.com/willowepublishing